
Executive Summary 
 
I work as an Accounts Payable clerk in the BYU Bookstore. There are many vendors with which the 
bookstore is in a credit balance. The AP department keeps track of all the monthly credit balances 
throughout the year. At the beginning of each month, one of the tasks is to download the end of month 
data from the bookstore’s accounting software. This data is then inputted into a summary file that 
shows the vendors and the credit balances throughout the year. This can be a challenge because first, 
the vendors with a zero balance during the month do not have a balance in the downloaded data, yet 
they exist in the year summary file. Second, some credit balances are new; vendors which previously did 
not have a credit balance now do. However, they do not exist in the year summary file. It is a tedious 
process to find which vendors exist and add rows of data to the file in the correct location. Previously, it 
would take my supervisor many valuable hours to input this data. My project saves her this time and 
helps her to avoid this menial task. 
 
My project performs the following key steps: 

 Creates a connection to the downloaded text file and imports the data to a new sheet in the file. 

 Formats the data for improved readability and enhanced transferability to the summary sheet. 

 Transfers the monthly data to the correct month location in the summary file based on the 
modified date of the downloaded text file. 

 Adds rows of data for the new vendors as needed. 

 Moves and saves the month data to the same directory as the year summary file. 
 
At the end of the year, a new file will need to be created for 2013. When the end of month file is 
downloaded next year and the program is run, it will generate a new file, clear out the old balances, 
create a new folder for the next year, and save the file in that new year folder. Thus, this program will 
continue for as long as the bookstore stands. 
 
Implementation Documentation 
 
Note1: It is very fortunate that when the accounting software downloads the end of month data, it 
automatically saves the data to the same location on the computer every time—C:\dell\excel.txt. This 
way, the program can always read the same file path. However, for the purposes of this demonstration, 
I have saved the file in another location—C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final 
Project\Copy of credit balance.txt. 
 
***Step 1: Determine what month we are planning to input. 
The first step is to determine what date the text file was downloaded. This will help us determine what 
month we are inputting. The AP department has a strict policy of not downloading end of month data 
until at least the first day of the next month. Thus, the date of the download will always correspond to 
the previous month’s data. For example, if the file was downloaded anytime in May, the month of the 
data corresponds to April. 
 
 
 



 
 
In the preceding screenshot, the variable LastModified returns the date and time of the last modification 
of the text file. Using the Month(LastModified) function, we can determine the number of the month. 
This is stored as the variable MonthNum. To determine the name of the corresponding month number, I 
created an array from 1 to 12 called WhatMonth. One challenge is what happens if the file was modified 
in January. One minus the month number of January (1 – 1) is equal to zero. The If statement in the code 
addresses this challenge. If the file was downloaded in January, then the month number is automatically 
assigned as 12 (December) and the year number (defined as the variable YearNum) is the previous year 
from the download year. Otherwise, the month is simply the previous month and the year is simply the 
same year. 
 
***Step 2: Obtain user’s confirmation. 
We don’t want the user to accidentally run the program without previously downloading the file from 
the accounting software (which automatically saves the data as a text file as explained in Note1). To 
address this, I created a series of message boxes to appear. The first box (on the left) gives the date of 
the text file download and asks for confirmation that the user is attempting. 
 

 
 
Upon selecting “Yes”, the program will continue to run. If “No” is selected, then another message box 
(on the right) will appear, indicating to the user to wait until after the first of the month to run the 
program. 
 



These message boxes were created using the following code: 
 
result = MsgBox("The last file was downloaded from VR on " & LastModified & ". Are you trying to input " & MonthName & " " 
& YearNum & "?", vbYesNo, "Confirmation") 
 
If result = vbNo Then 
MsgBox "Please download after the first of the month." 
Exit Sub 
End If 
 

I also created a message box that will appear and cause the sub procedure to close if there already exists 
a saved file for that specific month. The code for this message box is: 
 
Dim filename As String 
filename = ThisWorkbook.Path 
If Dir(filename & "\" & MonthName & YearNum & ".xlsm") <> vbNullString Then 
MsgBox "Error. This month has already been inputted. Please wait until next month." 
Exit Sub 
End If 
 

Note2: If you are following the code that appears in Exhibit A, you will notice that the next part of the 
code executes according to the year (If it is a new year, a new file will be created). I will address that 
challenge forthcoming. 
 
***Step 3: Make a connection to text file. 
The next step is to add a new sheet and create a connection in order to import the text file into the 
sheet. This is done by the following code: 
 
Sheets.Add 
     
    With ActiveSheet.QueryTables.Add(Connection:= _ 
        "TEXT;C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\Copy of credit balance.txt", _ 
        Destination:=Range("$A$1")) 
        .TextFileColumnDataTypes = Array(2, 2, 2, 1) 
        .Refresh BackgroundQuery:=False 
    End With 
      
  ActiveWorkbook.Connections("Copy of credit balance").Delete 
   
  ActiveSheet.Name = "CB" 

 

The code also deletes the connection after importing the data to ensure that the data does not get 
“refreshed” or changed. Then the code changes the name of the sheet to “CB”. The following screenshot 
shows the data in the text file and then the data after it has been imported into a spreadsheet in the 
workbook. 
 



 
 
 
 

 
 
***Step 4: Format the imported data. 
There exist various flaws in the imported data. First, there are additional spaces within each cell that 
must be trimmed. Also, if the vendor is longer than a specified length, then parts of the vendor get cut 
off and moved to the next row.  We need to concatenate that portion of the name and delete the blank 
rows. We also need to delete the extra vendor names that exist in column two. Additionally, we need to 
sort the data to make it easier to transfer into the year summary sheet. Finally, the heading will be 
renamed to reflect the month and year of the data. These types of formatting require relatively simple 
code: 
 
Dim FinalRow As Long 
Dim Col As Long 



FinalRow = Sheets("CB").Cells(Rows.Count, 1).End(xlUp).Row 
 
For Row = 1 To FinalRow 
 
    For Col = 1 To 4 
    Cells(Row, Col).Value = Trim(Cells(Row, Col).Value) 
    Next 
 
 
        If Cells(Row, 1).Value = "" And Cells(Row, 2).Value = "" And Cells(Row, 4).Value = "" Then 
        Cells(Row - 1, 3).Value = Cells(Row - 1, 3).Value & " " & Cells(Row, 3).Value 
        End If 
Next 
         
Cells(2, 2).EntireColumn.Delete 
 
 
For Row = FinalRow To 2 Step -1 

If Cells(Row, 1).Value = "" And Cells(Row, 3).Value = "" Then 
Cells(Row, 1).EntireRow.Delete 
End If 

Next 
 
Columns("A:C").EntireColumn.AutoFit 
 
    Range("A2").Select 
    Range(Selection, Selection.End(xlToRight)).Select 
    Range(Selection, Selection.End(xlDown)).Select 
    Sheets("CB").Sort.SortFields.Clear 
    Sheets("CB").Sort.SortFields.Add Key:=Range(Cells(3, 1), Cells(3, 1).End(xlDown)) _ 
        , SortOn:=xlSortOnValues, Order:=xlAscending 
       
      With Sheets("CB").Sort 
          .SetRange Selection 
          .Header = xlYes 
          .MatchCase = False 
          .Orientation = xlTopToBottom 
          .SortMethod = xlPinYin 
          .Apply 
      End With 
     
FinalRow = Sheets("CB").Cells(Rows.Count, 1).End(xlUp).Row 
 
    Range(Cells(3, 1), Cells(FinalRow, 1)).Select 
    Cells(3, 7).Value = 1 
    Cells(3, 7).Copy 
    Selection.PasteSpecial Paste:=xlPasteAll, Operation:=xlMultiply 
    Application.CutCopyMode = False 
    Cells(3, 7).ClearContents 
    Selection.HorizontalAlignment = xlLeft 
 
 
    Range(Cells(3, 3), Cells(FinalRow, 3)).Select 
    Selection.NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
 
 
Sheets("CB").Cells(1, 1).Value = MonthName & " " & YearNum & " Credit Balances" 

***Step 5: Transfer data to year summary file. 
Now that the month data has been formatted, it is ready to be entered into the year summary file. The 
two sheets look like this: 
 
Month Data 

 
 
Year Summary 

 
 

 As you can see above, many vendors in the year summary do not appear in the month data because 
they have a credit balance of zero. Also, there exist vendors in the month data that do not appear in the 



year summary sheet. For example, vendor number 1335 “BOXERCRAFT INCORPORATED” does not exist 
in the year summary file and a row will need to be inserted between vendor number 1284 and vendor 
number 1340 in the year summary sheet where that new vendor can be inputted. In order to address 
these two concerns, I designed a series of loops—one outside loop with two loops inside. 
 
The outside loop goes through each row of data on the month sheet (the data to be transferred). The 
first inside loop selects each row in the year summary sheet. The purpose of the inside loop is to find a 
vendor match in the year summary sheet.  Within the first inside loop, if the value in column a of the 
month sheet equals the value of column a of the year summary sheet, then the value in the relative 
month in the year file (measured by MonthNum + 2 columns) is equal to the value in column c (the 
credit balance amount) of the month file. The next step is to note on the month file next to the credit 
balance that this vendor and balance existed previously in the year file (“Existing Balance” will appear in 
column e). Also, if the statement is true, then the start x value (where to start on the row for the inside 
loop) increases by one. That way the loop does not have to return and loop through rows that have 
already been inputted. If not for this, then the first inside loop would start at 3 for every single row in 
the outside loop. The loop also exits so that the loop does not continue pointlessly. 
 
The second inside loop occurs when the first inside loop went unsuccessfully through each row of the 
year file and was not able to find a vendor match. The way that this second loop knows that the loop did 
not find a vendor match is that there will be no “Existing Balance” appearing in column e of the month 
file. This will signal the start of the second loop—the insert vendor loop. Now, we want to find the exact 
row where the vendor number fits (i.e. the vendor number is greater than the row above it and less than 
the row below it). To accomplish this, the If statement will be used to find the row where the vendor 
number ceases to be greater than the other rows. The second inside loop will insert a row, insert the 
vendor number in column a, insert the vendor name in column b, and insert the credit balance in the 
appropriate column (measured by MonthNum + 2). Then, like the first loop, the code will make a note 
on the month sheet (“New Balance” will appear in column e). This time, the note means that this vendor 
had a credit balance of zero previously and is a new credit balance. Also like the first inside loop, the 
code will add one to the start x value so that the loop does not start at the top of the year summary 
sheet again. 
 

 

New Month Sheet 



 
 

The code to transfer month data to the year summary sheet: 
 
Col = MonthNum + 2 
Dim x As Long 
x = 3 
 
 
For Row = 3 To FinalRow 
 
 
    For BKSTRow = x To FinalRowBKSTData 
        If Sheets("CB").Cells(Row, 1).Value = Sheets("BKST Data").Cells(BKSTRow, 1).Value Then             
            Sheets("BKST Data").Cells(BKSTRow, Col).Value = Sheets("CB").Cells(Row, 3).Value 
            Sheets("BKST Data").Cells(BKSTRow, Col).NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
            Sheets("CB").Cells(Row, 5).Value = "Existing Balance" 
            x = BKSTRow + 1 
            Exit For 
        End If 
    Next 
 
 
    If Sheets("CB").Cells(Row, 5).Value = "" Then 
        For BKSTRow = x To FinalRowBKSTData 
            If Sheets("BKST Data").Cells(BKSTRow, 1).Value > Sheets("CB").Cells(Row, 1).Value Then 
                Sheets("BKST Data").Cells(BKSTRow, 1).EntireRow.Insert 
                Sheets("BKST Data").Cells(BKSTRow, Col).Value = Sheets("CB").Cells(Row, 3).Value 
                Sheets("BKST Data").Cells(BKSTRow, Col).NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
                Sheets("BKST Data").Cells(BKSTRow, 1).Value = Sheets("CB").Cells(Row, 1).Value 
                Sheets("BKST Data").Cells(BKSTRow, 2).Value = Sheets("CB").Cells(Row, 2).Value 
                Sheets("CB").Cells(Row, 5).Value = "New Balance" 
                x = BKSTRow + 1 
                FinalRowBKSTData = Sheets("BKST Data").Cells(Rows.Count, 1).End(xlUp).Row 
                Exit For 
            End If 
        Next 
    End If 
Next 

New Year Summary Sheet 

Outside Loop 

Inside Loop #1 

Inside Loop #2 See discussion on 

previous page 



 

***Step 6: Move month data to new workbook, save in same directory, and save year summary. 
After looping through all the rows of data on the month sheet, the update of the year summary sheet is 
complete. The month sheet is ready to be moved to a new workbook and saved in the same directory as 
backup. The year summary will then be saved. The following code executes this step: 
 
Sheets("CB").Move 
 
ChDir filename 
 
Sheets("CB").Name = MonthName & " Credit Balance" 
 
ActiveWorkbook.SaveAs filename:=filename & "\" & MonthName & YearNum & ".xlsm", 
FileFormat:=xlOpenXMLWorkbookMacroEnabled 
 
ActiveWorkbook.Close 
 
ActiveWorkbook.Save 
 
Application.Quit 

 
***Step7: Create new file when next year has arrived. 
Note2 states that between step 2 and step 3, additional code determines whether a new file needs to be 
created for the next year. We will now return to discuss this step. 
 
First, if the year (YearNum) is not equal to the year noted in Cell A1 of the year summary sheet, then this 
month data does not belong in this year’s sheet and a new file needs to be created (Note: this will occur 
in February of each year). A message box will appear to confirm the user’s intention. Then, a new folder 
will be created to house the new year file using the function MkDir. Next, we need to use the SaveAs 
function instead of the SaveCopyAs function. If we use the SaveCopyAs function, then the previous year 
summary file will remain the active workbook. Using the SaveAs function will be equivalent of making a 
copy of the current file in a new folder location. In addition, this will allow us to maintain VBA code 
across different years of files. 
 
Dim NewYear As Boolean 
NewYear = False 
 
If YearNum <> Cells(1, 1).Value Then 
result2 = MsgBox("This will create a new file for " & YearNum & ". Is this correct?", vbYesNo, "Are you sure?") 
    If result2 = vbYes Then 
    NewYear = True 
 
    On Error Resume Next 
    MkDir "C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\" & YearNum 
    On Error GoTo 0 
     
    ActiveWorkbook.Save 
    ActiveWorkbook.SaveAs filename:="C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\" _ 
    & YearNum & "\AP Credit Blances" & YearNum & ".xlsm", FileFormat:=xlOpenXMLWorkbookMacroEnabled 
     
    Else 
    Exit Sub 
    End If 
End If 
 

Moves the month data sheet to a new workbook. This workbook 

Renames the sheet 

This following code is then executed to format the new year sheet. We first need to change the year of 
the file in Cell A1. Then, the code deletes all rows except for two. These will be the lower and upper 
bounds of the vendor numbers so that when the loops are executed to bring in the month data, the 
rows will be inserted between these bounds. After the loops are executed, the final code at the bottom 
is executed. This deletes those lower and upper bounds.  
 
Dim Row As Long 
Dim BKSTRow As Long 
Dim FinalRowBKSTData As Long 
 
If NewYear = True Then 
     
    Cells(1, 1).Value = YearNum 
     
    FinalRowBKSTData = Sheets("BKST Data").Cells(Rows.Count, 1).End(xlUp).Row 
     
        For BKSTRow = FinalRowBKSTData To 5 Step -1 
     
        Sheets("BKST Data").Cells(BKSTRow, 1).EntireRow.Delete 
     
        Next 
     
    Cells(3, 1).Value = "0" 
     
    Cells(4, 1).Value = "99999" 
 
End If 
 
 
 
 
 
If NewYear = True Then 
 
    Sheets("BKST Data").Cells(FinalRowBKSTData, 1).EntireRow.Delete 
 
    Sheets("BKST Data").Cells(3, 1).EntireRow.Delete 
 
End If 

 
 
Discussion of Learning and Difficulties Encountered 
 
This project was very fulfilling. It challenged my understanding of Visual Basic and I was really tested in 
creating a solution for this business problem. The main difficulty I encountered was the code created for 
step 5—inputting the month data into the year sheet. Many loops were performed that inserted rows 
and vendors in wrong locations before I developed a loop that would correctly input the data from the 
month sheet to the year sheet. One possibility that I considered in solving step 5 was to simply skip the 
new vendors and insert rows at the end of the year data. Then, I considered using the sort function to 
return the vendors to ascending order. Conceptually, I did not believe that doing it that way was as 
efficient as using two loops inside of another loop. I believe that using the loops is concise and more 
professional. I also experienced a little difficulty creating a new workbook for the next year. The 
challenge was how to save the file to a new folder while also maintaining the VBA code and keeping 
open the newly created file. I believe that the way I addressed the problem was the correct one. 

Assistance 
 
I did not receive substantial assistance from another person or persons. I did receive some help via 
online searches of the problems I was encountering. I also received semester-long tutelage from the 
amazing Professor Gove Allen. 
 
Exhibit A: Full VBA Code 
 
Sub callGetData(control As IRibbonControl) 
    Call InputData 
End Sub 
 
Sub InputData() 
 
Dim LastModified As Date 
LastModified = FileDateTime("C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\Copy of credit balance.txt") 
 
Dim WhatMonth(1 To 12) As String 
Dim MonthName As Variant 
Dim MonthNum As Integer 
Dim YearNum As Integer 
 
WhatMonth(1) = "January" 
WhatMonth(2) = "February" 
WhatMonth(3) = "March" 
WhatMonth(4) = "April" 
WhatMonth(5) = "May" 
WhatMonth(6) = "June" 
WhatMonth(7) = "July" 
WhatMonth(8) = "August" 
WhatMonth(9) = "September" 
WhatMonth(10) = "October" 
WhatMonth(11) = "November" 
WhatMonth(12) = "December" 
 
If Month(LastModified) = 1 Then 
MonthNum = 12 
YearNum = Year(LastModified) - 1 
Else 
MonthNum = Month(LastModified) - 1 
YearNum = Year(LastModified) 
End If 
 
MonthName = WhatMonth(MonthNum) 
 
result = MsgBox("The last file was downloaded from VR on " & LastModified & ". Are you trying to input " _ 
& MonthName & " " & YearNum & "?", vbYesNo, "Confirmation") 
 
If result = vbNo Then 
 
MsgBox "Please download after the first of the month." 
 
Exit Sub 
 
End If 
 
Dim filename As String 



 
filename = ThisWorkbook.Path 
 
If Dir(filename & "\" & MonthName & YearNum & ".xlsm") <> vbNullString Then 
 
MsgBox "Error. This month has already been inputted. Please wait until next month." 
 
Exit Sub 
 
End If 
 
 
Dim NewYear As Boolean 
NewYear = False 
 
If YearNum <> Cells(1, 1).Value Then 
result2 = MsgBox("This will create a new file for " & YearNum & ". Is this correct?", vbYesNo, "Are you sure?") 
    If result2 = vbYes Then 
    NewYear = True 
 
    On Error Resume Next 
    MkDir "C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\" & YearNum 
    On Error GoTo 0 
     
    ActiveWorkbook.Save 
    ActiveWorkbook.SaveAs filename:="C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\" _ 
    & YearNum & "\AP Credit Blances" & YearNum & ".xlsm", FileFormat:=xlOpenXMLWorkbookMacroEnabled 
     
    Else 
    Exit Sub 
    End If 
End If 
 
Dim Row As Long 
Dim BKSTRow As Long 
Dim FinalRowBKSTData As Long 
 
If NewYear = True Then 
     
    Cells(1, 1).Value = YearNum 
     
    FinalRowBKSTData = Sheets("BKST Data").Cells(Rows.Count, 1).End(xlUp).Row 
     
        For BKSTRow = FinalRowBKSTData To 5 Step -1 
     
        Sheets("BKST Data").Cells(BKSTRow, 1).EntireRow.Delete 
     
        Next 
 
    Cells(3, 1).Value = "0" 
    Cells(4, 1).Value = "99999" 
 
End If 
 
Sheets.Add 
     
    With ActiveSheet.QueryTables.Add(Connection:= _ 
        "TEXT;C:\Users\David\Documents\.........Fall 2012\VBA EXCEL\Final Project\Copy of credit balance.txt", _ 



        Destination:=Range("$A$1")) 
        .TextFileColumnDataTypes = Array(2, 2, 2, 1) 
        .Refresh BackgroundQuery:=False 
    End With 
     
    Range("C5").Select 
   
  ActiveWorkbook.Connections("Copy of credit balance").Delete 
   
  ActiveSheet.Name = "CB" 
 
Dim FinalRow As Long 
Dim Col As Long 
 
FinalRow = Sheets("CB").Cells(Rows.Count, 1).End(xlUp).Row 
 
For Row = 1 To FinalRow 
     
    For Col = 1 To 4 
     
    Cells(Row, Col).Value = Trim(Cells(Row, Col).Value) 
     
    Next 
 
        If Cells(Row, 1).Value = "" And Cells(Row, 2).Value = "" And Cells(Row, 4).Value = "" Then 
 
        Cells(Row - 1, 3).Value = Cells(Row - 1, 3).Value & " " & Cells(Row, 3).Value 
 
        End If 
 
Next 
         
Cells(2, 2).EntireColumn.Delete 
 
For Row = FinalRow To 2 Step -1 
 
If Cells(Row, 1).Value = "" And Cells(Row, 3).Value = "" Then 
 
Cells(Row, 1).EntireRow.Delete 
 
End If 
 
Next 
 
Columns("A:C").EntireColumn.AutoFit 
 
    Range("A2").Select 
    Range(Selection, Selection.End(xlToRight)).Select 
    Range(Selection, Selection.End(xlDown)).Select 
    Sheets("CB").Sort.SortFields.Clear 
    Sheets("CB").Sort.SortFields.Add Key:=Range(Cells(3, 1), Cells(3, 1).End(xlDown)) _ 
        , SortOn:=xlSortOnValues, Order:=xlAscending 
       
      With Sheets("CB").Sort 
          .SetRange Selection 
          .Header = xlYes 
          .MatchCase = False 
          .Orientation = xlTopToBottom 



          .SortMethod = xlPinYin 
          .Apply 
      End With 
       
FinalRow = Sheets("CB").Cells(Rows.Count, 1).End(xlUp).Row 
 
    Range(Cells(3, 1), Cells(FinalRow, 1)).Select 
    Cells(3, 7).Value = 1 
    Cells(3, 7).Copy 
    Selection.PasteSpecial Paste:=xlPasteAll, Operation:=xlMultiply 
    Application.CutCopyMode = False 
    Cells(3, 7).ClearContents 
    Selection.HorizontalAlignment = xlLeft 
    Range(Cells(3, 3), Cells(FinalRow, 3)).Select 
    Selection.NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
 
Sheets("CB").Cells(1, 1).Value = MonthName & " " & YearNum & " Credit Balances" 
 
FinalRowBKSTData = Sheets("BKST Data").Cells(Rows.Count, 1).End(xlUp).Row 
 
Col = MonthNum + 2 
 
Dim x As Long 
 
x = 3 
For Row = 3 To FinalRow 
 
    For BKSTRow = x To FinalRowBKSTData 
        If Sheets("CB").Cells(Row, 1).Value = Sheets("BKST Data").Cells(BKSTRow, 1).Value Then 
             
            Sheets("BKST Data").Cells(BKSTRow, Col).Value = Sheets("CB").Cells(Row, 3).Value 
             
            Sheets("BKST Data").Cells(BKSTRow, Col).NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
             
            Sheets("CB").Cells(Row, 5).Value = "Existing Balance" 
             
            x = BKSTRow + 1 
             
            Exit For 
        End If 
    Next 
 
    If Sheets("CB").Cells(Row, 5).Value = "" Then 
        For BKSTRow = x To FinalRowBKSTData 
            If Sheets("BKST Data").Cells(BKSTRow, 1).Value > Sheets("CB").Cells(Row, 1).Value Then 
                 
                Sheets("BKST Data").Cells(BKSTRow, 1).EntireRow.Insert 
                 
                Sheets("BKST Data").Cells(BKSTRow, Col).Value = Sheets("CB").Cells(Row, 3).Value 
                 
                Sheets("BKST Data").Cells(BKSTRow, Col).NumberFormat = "#,##0.00_);[Red](#,##0.00)" 
                 
                Sheets("BKST Data").Cells(BKSTRow, 1).Value = Sheets("CB").Cells(Row, 1).Value 
                 
                Sheets("BKST Data").Cells(BKSTRow, 2).Value = Sheets("CB").Cells(Row, 2).Value 
                 
                Sheets("CB").Cells(Row, 5).Value = "New Balance" 
                 



                x = BKSTRow + 1 
                 
                FinalRowBKSTData = Sheets("BKST Data").Cells(Rows.Count, 1).End(xlUp).Row 
                 
                Exit For 
            End If 
        Next 
 
End If 
Next 
 
If NewYear = True Then 
 
    Sheets("BKST Data").Cells(FinalRowBKSTData, 1).EntireRow.Delete 
 
    Sheets("BKST Data").Cells(3, 1).EntireRow.Delete 
 
End If 
 
Sheets("CB").Move 
 
ChDir filename 
 
Sheets("CB").Name = MonthName & " Credit Balance" 
 
ActiveWorkbook.SaveAs filename:=filename & "\" & MonthName & YearNum & ".xlsm", 
FileFormat:=xlOpenXMLWorkbookMacroEnabled 
 
ActiveWorkbook.Close 
 
ActiveWorkbook.Save 
 
Application.Quit 
 
End Sub 


